
Abstract—We introduce a novel methodology to implement the
physiological transition between tonic spiking and bursting in
electronic circuits composed of resistors, capacitors and transis-
tors. The result is a six-transistor neuromorphic device organized
by the same geometry of high-dimensional electrophysiological
neuron models and therefore exhibiting the same qualitative
behavior.

I. INTRODUCTION

Nature offers spectacular examples of energy-efficient,
lightweight control architectures. Flight control in a sim-
ple animal like a honeybee outperforms the latest robotic
architectures in terms of energy consumption, adaptability,
robustness, and dimensions. Neuromorphic engineering aims
at emulating the way in which biological neuronal systems
perceive and represent the outside world, take decisions and
develop computations, and command motor outputs [1], [2].

In implementing the dynamical behavior of biological neu-
rons in electronic hardware we face the compromise between
fidelity of the reproduced behavior and complexity of the de-
signed circuit. Existing silicon neuron designs span a variety of
solutions: from detailed implementation of neuron biophysics
[3] to implementation of simple, abstract neuron models [4].
Both approaches have advantages and disadvantages, and it is
an active research area to determine which implementation to
use depending on the desired objective [5].

The possibility of reliably and rapidly switching between
distinct dynamical modes is one of the peculiarity of biolog-
ical neurons, which allows them to adapt their input-output
response depending on internal and environmental conditions.
Two fundamental neuronal activity modes are tonic spiking
and bursting. Tonic spiking describes the slow, regular gen-
eration of spikes in the neuron membrane potential. Bursting
describes the alternation between moments of low membrane
potential and moments of high oscillatory activity, in which
spikes are generated at very high frequency. The transition
between tonic spiking and bursting plays a major role in the
processing of sensory information [6], [7].

We showed [8], [9] that all biological neurons share the
same geometry for the transition between tonic spiking and
bursting. In particular, this transition can be described in a
simple, abstract model given by the normal form of an orga-
nizing singularity. Roughly speaking, a singularity describes
a highly degenerate and fragile situation that correspond

to the transition between distinct regimes [10]. There is a
direct correspondence between biophyisical parameters and
mathematical parameters in the abstract model, which leads
to a profound mathematical understanding of robustness and
modulation of neuronal activity [11]. We further showed that
the same qualitative picture can be realized in simple circuits
[12].

In this paper, we follow the recipe provided in [12] to design
a neuromorphic circuit with the property of sharing the same
qualitative geometry, robustness, and modulation capabilities
of biophysical neuron models. As a first, biologically relevant
exploration, we focus on the transition between tonic spiking
and bursting. The resulting circuit solely uses six transistors
and passive elements. For comparison, the simplest available
neuromorphic circuit capable of transitioning between tonic
spiking and bursting uses fourteen mosfet transistors [4].
Another advantage of our approach is that circuit parameters
are constructively tuned by geometric inspection of its static
input-output characteristic, which avoids laborious and non-
constructive parameter fitting procedures.

In Section II we rapidly review the results in [8], [9], [12].
Grounded in these works, we derive an implementation of our
neuromorphic device in Section III. The appendix contains the
necessary code for simulating it.

II. THE GEOMETRY OF NEURONAL BURSTING AND ITS
BLOCK REALIZATION

Electrophysiological models of neurons are constructed
upon the seminal work of Hodgkin and Huxley [13]. They
all share the physical interpretation of the nonlinear RC
circuit depicted in Fig. 1-A. The capacitor models the neuron
membrane and the other branches, containing a voltage source
and a variable resistance, model the flow of a specific ion
across the membrane.

Ion flow across the membrane is dynamically regulated by
the membrane potential via opening and closing of the ion
channels, which makes the circuit in Fig. 1-A highly nonlinear.
As such, it can exhibit a rich variety of dynamical behaviors.
The present paper focuses on two fundamentals behaviors
shared by almost all neuron types: the tonic spiking behavior
of Fig. 1-B left and the bursting behavior of Fig. 1-B right.

Reproducing tonic spiking and bursting, as well as the
transition between these two modes, in an electrophysiolog-

Fig. 1: A. The RC circuit associated to a conductance-based
biophysical model of neuron. B. Two dynamical modes in a
conductance-based model. Left: tonic spiking. Right: bursting.
Membrane potential values are in millivolts. Time is in mil-
lisecond.

ical model requires a fine tuning of the many biophysical
parameters that usually ends up in an extensive brute-force
computational parameter search [14]. A different approach
relies on bifurcation theory [15].

Roughly speaking, bifurcation theory makes the ansatz that
the vector field associated to an electrophysiological model
undergoes some qualitative change at the transition between
two distinct dynamical modes.

We showed in [8], [9] that the bifurcation associated to
the transition between tonic spiking and bursting can be
algebraically tracked by exploiting the multi-timescale nature
of electrophysiological neuron models and by detecting a tran-
scritical singularity in the critical manifold of the associated
singularly perturbed dynamics. We refer the reader to [16] for
an introduction to geometric singular perturbation theory and
to [10] for singularity theory concepts.

The power of this analysis is that we can visualize the
geometry of the tonic spiking - bursting transition in a low-
dimensional normal form of the organizing singularity:

ẋ = −x3 − (λ+ y)2 + βx− α− z (1a)
ẏ = εs(x− y) (1b)
ż = εu(x− z), (1c)

where λ is called the bifurcation parameter, α, β are called
unfolding parameters, and 0 < εu � εs � 1 model timescale
separation between the three state variables x, y and z. The
distinction between bifurcation and unfolding parameters is
instrumental to the tools used in the construction of the normal
form (1), that is, singularity theory applied to bifurcation
problems [10].

Fig. 2-A shows the temporal traces and the projection onto
the phase plane of the slow-fast subsystem (1a-1b) of tonic
spiking and bursting behaviors in model (1). It shows that,

Fig. 2: A. Top: temporal traces of mode (1) in tonic spiking
and bursting. Bottom: projection of the trajectory on the x−y
phase plane. The trajectory is in red. The critical manifold
(2) is black for small value of z and in gray for large value
of z. B. The mirrored hysteresis bifurcation diagram. C. The
geometry of tonic spiking and bursting. The black dot denotes
stable fixed point, the circle unstable fixed points. Stable limit
cycles are drawn in blue.

due to timescale separation, trajectories spend most of the time
near the critical manifold

Z := {(x, y, z) ∈ R3 : ẋ = 0}, (2)

that is, the x-nullcline composed of x steady states as y and
z varies.

The fundamental shape of the critical manifold underly-
ing the transition between tonic spiking and bursting is the
mirrored hysteresis bifurcation diagram introduced in [9] and
sketched in Fig. 2-B. In the tonic spiking mode trajectories
solely visits the right branch of the mirrored hysteresis,
whereas in bursting mode trajectories alternate between the
two branches.

The geometry of both behaviors is sketched in Fig. 2-C. In
tonic mode, for each value of the ultra-slow variable z, the
slow-fast subsystem possesses a single attractor on the right
branch of the mirrored hysteresis: either a stable fixed point,
for large value of z (left plot), or a stable limit cycle, for small
value of z (right plot). For large initial values of z the model
is therefore at quasi-steady state. The ultraslow dynamics (1c)
lets in this case z decrease until the steady state loses stability
and a spike is emitted along the newborn limit cycle. This in
turn leads to a sharp increase of z, which immediately lets the
steady state recover stability, and the model is forced back to
quasi-steady state.

In bursting mode, there exists a large range of values of
the ultraslow variable z in which the slow-fast subsystem
exhibits bistability between a stable steady state on the left
branch of the mirrored hysteresis and a limit cycle on the
right branch of the mirrored hysteresis (center plot). For large

Fig. 3: Block realization of a mirrored hysteresis nonlinear
characteristic.

initial value of z the sole attractor is a stable steady state (left
plot). The ultraslow dynamics (1c) lets in this case z decrease.
However, once in the bistable region, the model remains at
quasi-steady state. Only for sufficiently large z the steady state
loses stability and the trajectory converges toward the spiking
limit cycle, which is now the sole attractor (left plot). On the
limit cycle (1c) lets z decrease. Again, all through the bistable
region the system remains in the oscillatory mode and only for
z sufficiently large the trajectory converges to its quasi-steady
state.

Normal forms are useful not only because they unmask
the geometry underlying a given dynamical behavior, but also
because they possess the minimum number of parameters to
reproduce a family of behaviors of interest. As such they can
also be implemented in physical devices much more easily and
robustly than the original biophysical model, yet, preserving
the same geometric and input/output properties.

We showed in [12] that the mirrored hysteresis bifurcation
diagram can be realized in the input-output diagram of Fig. 3.
Its basic ingredients are a non-monotone nonlinearity cascaded
with a saturation nonlinearity and a positive feedback loop
around the saturation nonlinearity. The positive feedback loop
transforms the saturation into a hysteresis [12, Proposition
1], the non-monotone block creates a mirror of the resulting
hysteretic characteristic.

Adding linear dynamical systems evolving on three sharply
different timescales (Fig. 4) transforms the static circuit in
Fig. 3 into a three-timescale dynamical system with the same
qualitative behavior of model (1). In particular, the circuit in
Fig. 4 exhibits the same geometric transition between tonic
spiking and bursting as model (1).

III. ELECTRONIC IMPLEMENTATION

In this section we derive an electronic implementation of
the circuit in Fig. 4. The main active component is an npn
transistor. Its constitutive relations are given by Ebers-Moll
equations [17] but, to simplify the analysis, we will model the
transistor as a current source with current proportional to that
of a diode standing at the base. Furthermore, will regard the
diode as a perfect switch that opens whenever vBE ≥ 0.6 and
closes otherwise.

Fig. 4: Block realization of a neuron model. The linear filters
Hf,s,u(s) are first order filters with sharply separated cut-off
frequency: large for Hf , intermediate for Hs, and small for
Hu.

Fig. 5: Basic saturation. Common-emitter configuration (left).
Piecewise linear approximation vs. exact solutions (right).

A. Voltage-controlled non-monotone characteristic

The non-monotone block of the circuit in Fig. 4 is real-
ized as the difference of two monotone nonlinearities [12].
A common emitter npn transistor configuration serves as a
simple, natural saturation monotone nonlinearity (see Fig. 5).
In view of the transistor model described above, the vy–v1
characteristic takes the piecewise linear form

v1 = vcc − ProjS(g1(vy − 0.6)) , (3)

where
g1 =

βRC

RB + (β + 1)RE

is the voltage gain (the slope of the saturation), β ≈ 100 is
the transistor’s current gain and S1 = [0, vs] with

vs = (vcc − 0.1)
RC

RC +RE

the saturation voltage. Usually, one chooses RB � βRE so
that g1 ≈ RC

RE
, that is, so that the dependence of g1 on β is

negligible. The operator ProjS1
projects its argument into the

set S1, that is, ProjS1
(v) = argminw∈S1

‖v − w‖. For our
particular S1, the projection translates into

ProjS1
(v) =


0 if v ≤ 0

v if 0 ≤ v ≤ vs
vs if vs ≤ v

.

To asses the quality of our estimation, we choose a set of
parameters and compare (3) against the simulation obtained



Fig. 6: Non monotone characteristic. Parallel interconnection
of a saturation and a linear gain (left). Piecewise linear
approximation vs. exact solutions (right).

using ngspice (which implements Ebers-Moll equations).
The results are shown in Fig. 5.

The parallel interconnection achieving the non-monotone
behavior is shown in Fig. 6, left. Applying Kirchhoff’s laws
and the piecewise linear model for the transistor one obtains

v4 = g2(vcc − ProjS(g1(vy − 0.6))) + g3vy (4)

with

g2 =
RSRA2

RS (RA1
+RA2

) +RA1
RA2

g3 =
RSRA1

RS (RA1
+RA2

) +RA1
RA2

(see Fig. 6, right).
The characteristic (4) is non-monotone whenever strict

extrema are present. A necessary condition for the presence
of extrema is 0 ∈ ∂v4 [18, p. 70], where ∂v4 is the
subdifferential [18, p. 32] of v4 with respect to vy . For (4)
we have ∂v4 = g3 − g1g2ΨS1

(g1(vy − 0.6)), where

ΨS1(v) =


0 if v /∈ S1

[0, 1] if v ∈ ∂S1

1 if v ∈ intS1

with intS1 and ∂S1 the interior and the boundary of S1,
respectively. Thus, 0 ∈ ∂v4 implies g1 > g3/g2.

Let us compute some points of interest of the characteris-
tic (4) (the computation of such points is useful for choosing
the appropriate resistors):

v4(0) = g2vcc

v4(0.6) = g2vcc + g30.6 (a local maximum)

v4

(
vs
g1

)
= g2vcc −

(
g2 −

g3
g1

)
vs (a local minimum)

v4(vcc) = (g2 + g3)vcc − g2vs
(local extrema are found by solving 0 ∈ ∂v4(vy)).

The possibility to modulate the non-monotone characteristic
is achieved by cascading a differential amplifier as the one
shown in Fig. 7, left. Kirchhoff’s laws together with the
piecewise linear model give

v5 = vcc − ProjS2
(g4(v4 − 0.6)− g5(vz − 0.6)) , (5)

Fig. 7: The voltage-controlled non-monotone characteristic is
realized by cascading a differential amplifier (left) with a
fixed non-monotone characteristic (see Fig. 6). Input–output
response for different values of vz (right).

where

g4 =
R̄C1

(
R̄E +RB2

)
R̄E (RB1 +RB2) +RB1RB2

g5 =
R̄C1

R̄E

R̄E (RB1 +RB2) +RB1RB2

,

R̄C = (β + 1)RC and R̄E = (β + 1)RE . The voltage v4 is
the non-inverting input and vz the inverting one. The complete
block is shown in Fig. 11 and the input–output characteristic
is shown in Fig. 7, right.

B. Hysteretic characteristic

The hysteretic block is built as the positive feedback of a
basic saturation and a linear gain [12]. This is achieved at once
with a differential amplifier like the one shown in Fig. 7, left.
By letting v5 be the inverting input, v6 the non-inverting input
and vx the output, we obtain (cf. (5))

vx = vcc − ProjS3
(g6(v5 − 0.6)− g7(v6 − 0.6)) .

Positive feedback is then achieved simply by setting v6 =
vx, as shown in Fig. 11. This results in the piecewise linear
characteristic

F (v5, vx) = vx − vcc+
ProjS3

(g6(v5 − 0.6)− g7(vx − 0.6)) = 0 . (6)

It follows from the implicit function theorem [19, p. 256] that
a necessary condition for the existence of singular points is
0 ∈ ∂F (v5, vx) with the subdifferential taken with respect to
vx. For (6) we have

∂F (v5, vx) = 1− g7ΨS3
(g6(v5 − 0.6)− g7(vx − 0.6)) ,

so that 0 ∈ ∂F (v5, vx) implies g7 > 1. There are two points
of singularity. The first one is characterized by g6(v5−0.6) =
g7(vx−0.6) which, together with the condition F (v5, vx) = 0,
gives

vx = vcc and v5 =
g6
g7
vcc +

g6 − g7
g6

0.6 .



(a) vz = 3.8V . A saddle point, an unstable
and a stable node are present (red). A stable
node is present (blue). In both cases, almost
all trajectories converge to the stable node.

(b) vz = 4.1V . A saddle point, a stable limit
cycle around an unstable node and a stable
node coexist (red). The only attractor is a
stable node (blue).

(c) vz = 4.7V . A stable limit cycle around an
unstable node exists. Almost all trajectories
converge to the limit cycle (red and blue).

Fig. 8: Mirrored-hysteresis. Solid black lines correspond to the
x-steady states found by solving the circuit with ngspice.
Dotted lines are manually added to sketch the solutions not
found by the solver.Red and blue lines correspond to the
desired vy-nullclines.

The other point of singularity is determined by g6(v5−0.6) =
g7(vx − 0.6) + vs, which gives

vx = vcc − vs and v5 =
g7
g6
vcc +

g6 − g7
g6

0.6− g7 − 1

g6
vs .

C. Voltage-controlled mirrored hysteresis

The cascade of the controlled non-monotone block and the
hysteresis (see Fig. 11) produces the desired voltage-controlled
mirror hysteresis (see Figs. 8a-8c, black). This characteristic
is qualitatively equivalent to Figs. 2-C.

D. Burster

We now transform the static circuit in Fig. 11 into a dynamic
circuit exhibiting the same qualitative dynamics as model (1).

The parasitic capacitances of the transistors provide the fast
vx dynamics and set its corresponding time-scale (cf. Hf in

Fig. 9: Transition between a stable node (constant output)
and a stable limit cycle (oscillations). The transitions occur
at different values of vz , which indicates that the stable node
and the stable limit cycle coexist for some values of vz .

Fig. 10: Bursting (red) and spiking (blue).

Fig. 4). The voltage vx is fed back to vy by means of a resistive
voltage divider and a capacitor.The values of the resistors and
the capacitor determine the time-scale of the slow vy dynamics
as well as the slope of its nullcline (cf. Hs in Fig. 4).

Figs. 8a-8c confirm the qualitative equivalence of the circuit
in Fig. 11 and model (1). By sweeping vz we recover the
same qualitative phase portraits of Fig. 2-C left, which underlie
the behavior simulated in Fig. 9. A key ingredient in the
bursting behavior is the bistability of the limit cycle and the
node (Fig. 8b, red). The presence of this phenomenon can be
asserted by noting that the transition from the constant output
(the stable node) to the oscillating behavior (the limit cycle)
occurs at a higher value of vz than the one for the transition
from the oscillating behavior to the constant output.

Bursting is finally achieved by feeding vx back to vz through
the ultra slow filter Hu. This is again realized with a voltage
divider and a capacitor, but now the circuit’s time-constant is
chosen much larger. To ensure a robust operation, the output
of Hu is amplified so that vz exhibits a large swing. In fact
it is the complement of vx that is passed through an amplifier
with negative slope (this accounts for the sixth transistor). The
time response is shown in Fig. 10.

E. Spiker

Recall that the mirrored hysteresis captures both modes
of operation: bursting and tonic spiking. Geometrically, the
difference between the two behaviors is the locus of the stable
fixed point, as sketched in Fig. 2. In our circuit, we recover

Fig. 11: Circuit realizing the mirrored hysteresis. By suitably
changing vz , the vy–vx characteristic can adopt the three forms
portrayed on Figs. 8a-8c (cf. the right part of Fig. 2).

the same geometric picture by changing the slope of the vy-
nullcline via tuning of the resistance Ri1 and Ri2 . When
the slope of the vy nullcline is sufficiently large this line
solely intersects the left branch of the mirrored hysteresis
(see Figs. 8b and 8c, blue), thus destroying the possibility
of bistability underlying bursting. The model is in this case in
the tonic spiking mode shown in Fig. 10.

IV. CONCLUSION AND PERSPECTIVES

We have implemented the biological transition between
tonic spiking and bursting in a compact six-transistor model.
This model can be used as the building block of neuromor-
phic sensory and computing devices. Future work will aim
at further ameliorating the circuit, at an efficient hardware
implementation, and at the assembly of small, medium, and
large networks implementing different biological functions,
from central pattern generation to attention control.

REFERENCES

[1] K. Boahen, “Neuromorphic microchips,” Scientific American, vol. 292,
no. 5, pp. 56–63, 2005.

[2] S.-C. Liu and T. Delbruck, “Neuromorphic sensory systems,” Current
opinion in neurobiology, vol. 20, no. 3, pp. 288–295, 2010.

[3] K. M. Hynna and K. Boahen, “Thermodynamically equivalent sili-
con models of voltage-dependent ion channels,” Neural Computation,
vol. 19, no. 2, pp. 327–350, 2007.

[4] J. H. Wijekoon and P. Dudek, “Compact silicon neuron circuit with
spiking and bursting behaviour,” Neural Networks, vol. 21, no. 2, pp.
524–534, 2008.

[5] G. Indiveri, B. Linares-Barranco, T. J. Hamilton, A. Van Schaik,
R. Etienne-Cummings, T. Delbruck, S.-C. Liu, P. Dudek, P. Häfliger,

APPENDIX
Ngspice code for the complete circuit

* this is complete.cir file

* voltage resources
vcc 5 0 dc 5V

* non monotone
q1 2 3 4 2n2222bis
rC1 5 2 16k
rB1 3 1 100k
rE1 4 0 10k
Ra1 6 1 100k
Ra2 6 2 33k
Rs 6 0 220k

* non monotone modulation
q2 7 8 9 2n2222bis
q3 10 11 9 2n2222bis
rC2 7 5 4.7k
rC3 10 5 4.7k
rB2 8 6 1k
rB3 12 11 1.2k
rE2 9 0 470

* hysteresis
q4 13 14 15 2n2222bis
q5 16 17 15 2n2222bis
rC4 13 5 820
rC5 16 5 240
rB4 14 7 2.4k
rB5 17 13 6k
rE4 15 0 240

* vy-vx feedback loop
ri1 13 1 15k
ri2 1 0 47k

* Set ri2 to 34.5k for tonic spiking
ciF 1 0 22n

* vz-vx feedback loop
q6 12 18 19 2n2222bis
ro1 16 18 4.7k
ro2 18 0 4.7k
coF 18 0 4.7u
rC6 12 5 200
rE6 19 0 20
rbi 19 5 150

* model for a 2n2222 transistor
.model 2n2222bis npn (is=14.34f bf=255.9 vaf=74.03 ikf=.2847
+ ise=14.34f ne=1.307 br=6.092 ikr=0 isc=0 nc=2 rb=10
+ rc=1 cje=22.01p tf=411.1p cjc=7.306p tr=46.91n xtb=1.5
+ Xti=3 Eg=1.11 Mjc=.3416 Vjc=.75 Fc=.5 Mje=.377 Vje=.75
+ Itf=.6 Vtf=1.7 Xtf=3 )
.control
tran 1us 40ms
plot v(16) ylimit 0 5
.endc
.end


